Pensieve header: Mathematica notebook for A Perturbed Alexander Invariant.

In[ ]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\APerturbedAlexanderInvariant"];
```

tex

Two of the main reasons we like ρ_1 is that it is very easy to implement and even an unsophisticated implementation runs very fast. To highlight these points we include a full implementation here, a step-by-step run-through, and a demo run. We write in Mathematica~\cite{Wolfram:Mathematica}, and you can find the notebook displayed here at~\cite[APAI.nb]{Self}.

We start by loading the library \verb$KnotTheory`$~\cite{Bar-NatanMorrison:KnotTheory} (it is used here only for the list of knots that it contains, and to compute other invariants for comparisons). We also load minor conversion routine~\cite[Rot.nb / Rot.m]{Self} whose internal workings are simple and yet irrelevant here.

pdf

In[ ]:=
```
Once[<< KnotTheory`; << Rot.m];
```

pdf

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

pdf

```
Loading Rot.m from http://drorbn.net/APAI to compute rotation numbers.
```

tex

\needspace{50mm}
\subsection{The Program} This done, here is the full ρ_1 program:

pdf

In[ ]:=
```
R₁[s_, i_, j_] := s (g[j,i] (g[j+1,j] + g[j,j+1] - g[i,j]) - g[i,i] (g[j,j+1] - 1) - 1/2);
ρ[K_] := Module[{Cs, φ, n, A, s, i, j, k, Δ, G, ρ1},
  {Cs, φ} = Rot[K]; n = Length[Cs];
  A = IdentityMatrix[2 n + 1];
  Cases[Cs, {s_, i_, j_} :> (A[[{i, j}, {i + 1, j + 1}]] += ({{-T^s, T^s - 1}, {0, -1}}))];
  Δ = T^((-Total[φ] - Total[Cs[[All, 1]]])/2) Det[A];
  G = Inverse[A];
  ρ1 = Sum[R₁ @@ Cs[[k]], {k, 1, n}] - Sum[φ[[k]] (g[k,k] - 1/2), {k, 1, 2 n}];
  Factor@{Δ, Δ^2 ρ1 /. g[α_, β_] :> G[[α, β]]}];
```

tex

The program uses mostly the same symbols as the text, so even without any knowledge of Mathematica, the reader should be able to recognize at least formulas~\eqref{eq:A}, \eqref{eq:Delta}, and~\eqref{eq:rho1} within it. As a further hint we add that the variables \verb$Cs$ ends up storing the list of crossing in a knot K, where each crossing is stored as a triple $\{s,i,j\}$, where s, i, and j have the same meaning as in~\eqref{eq:A}. The conversion routine \verb$Rot$ automatically produces

\verb$Cs$, as well as a list φ of rotation numbers, given any other knot presentation known to the package \verb$KnotTheory`$.

tex

Note that the program outputs the ordered pair (Δ,ρ_1). The Alexander polynomial Δ is anyway computed internally, and we consider the aggregate (Δ,ρ_1) as more interesting than any of its pieces by itself.

tex

\subsection{A Step-by-Step Run-Through} We start by setting K to be the knot diagram on page~1 using the \verb$PD$ notation of \verb$KnotTheory`$. We then print \verb$Rot[K]$, which is a list of crossings followed by a list of rotation numbers:

pdf

```
In[∘]:= K = PD[X[4, 2, 5, 1], X[2, 6, 3, 5], X[6, 4, 7, 3]];
        Rot[K]
```

pdf

```
Out[∘]= {{{1, 1, 4}, {1, 5, 2}, {1, 3, 6}}, {0, 0, 0, -1, 0, 0}}
```

tex

Next we set \verb$Cs$ and φ to be the list of crossings, and the list of rotation numbers, respectively.

pdf

```
In[∘]:= {Cs, φ} = Rot[K]
```

pdf

```
Out[∘]= {{{1, 1, 4}, {1, 5, 2}, {1, 3, 6}}, {0, 0, 0, -1, 0, 0}}
```

tex

We set \verb$n$ to be the number of crossings, \verb$A$ to be the $(2n+1)$-dimensional identity matrix, and then we iterate over \verb$c$ in \verb$Cs$, adding a block as in~\eqref{eq:A} for each crossings.

pdf

```
In[∘]:= n = Length[Cs];
        A = IdentityMatrix[2 n + 1];
        Cases[Cs, {s_, i_, j_} :> (A[[{i, j}, {i + 1, j + 1}]] += ( -T^s  T^s - 1
                                                                     0    -1    ))];
```

tex

\needspace{40mm}
Here's what \verb$A$ comes out to be:

pdf

```
In[∘]:= A // MatrixForm
```

Out[∘]//MatrixForm=
pdf

$$\begin{pmatrix} 1 & -T & 0 & 0 & -1+T & 0 & 0 \\ 0 & 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & -T & 0 & 0 & -1+T \\ 0 & 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & -1+T & 0 & 1 & -T & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & -1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

Here's the same, in TeXForm:

In[∘]:=
```
A // MatrixForm // TeXForm
```

Out[∘]//TeXForm=
```
\left(
\begin{array}{ccccccc}
 1 & -T & 0 & 0 & T-1 & 0 & 0 \\
 0 & 1 & -1 & 0 & 0 & 0 & 0 \\
 0 & 0 & 1 & -T & 0 & 0 & T-1 \\
 0 & 0 & 0 & 1 & -1 & 0 & 0 \\
 0 & 0 & T-1 & 0 & 1 & -T & 0 \\
 0 & 0 & 0 & 0 & 0 & 1 & -1 \\
 0 & 0 & 0 & 0 & 0 & 0 & 1 \\
\end{array}
\right)
```

tex

We set \$\Delta\$ to be the determinant of \verb\$A\$, with a correction as in~\eqref{eq:Delta}. So \$\Delta\$ is the Alexander polynomial of \$K\$.

In[∘]:=
```
Det[A]
```

Out[∘]= $1 - T + T^2$

pdf

In[∘]:= $\Delta = T^{(-\text{Total}[\varphi]-\text{Total}[\text{Cs}[[\text{All},1]]])/2}\,\text{Det}[A]$

pdf

Out[∘]= $\frac{1 - T + T^2}{T}$

tex

\verb\$G\$ is now the \verb\$Inverse\$ of \verb\$A\$:

pdf

In[∘]:=
```
G = Inverse[A];
G // MatrixForm
```

Out[∘]//MatrixForm=
pdf

$$\begin{pmatrix} 1 & \frac{T-T^2+T^3}{1-T+T^2} & 1 & \frac{T-T^2+T^3}{1-T+T^2} & 1 & \frac{T-T^2+T^3}{1-T+T^2} & 1 \\ 0 & 1 & \frac{1}{1-T+T^2} & \frac{T}{1-T+T^2} & \frac{T}{1-T+T^2} & \frac{T^2}{1-T+T^2} & 1 \\ 0 & 0 & \frac{1}{1-T+T^2} & \frac{T}{1-T+T^2} & \frac{T}{1-T+T^2} & \frac{T^2}{1-T+T^2} & 1 \\ 0 & 0 & \frac{1-T}{1-T+T^2} & \frac{1}{1-T+T^2} & \frac{1}{1-T+T^2} & \frac{T}{1-T+T^2} & 1 \\ 0 & 0 & \frac{1-T}{1-T+T^2} & \frac{T-T^2}{1-T+T^2} & \frac{1}{1-T+T^2} & \frac{T}{1-T+T^2} & 1 \\ 0 & 0 & 0 & 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

Here's the same, in TeXForm:

In[∘]:=
```
G // MatrixForm // TeXForm
```

Out[∘]//TeXForm=
```
\left(
\begin{array}{ccccccc}
 1 & \frac{T^3-T^2+T}{T^2-T+1} & 1 &
   \frac{T^3-T^2+T}{T^2-T+1} & 1 &
   \frac{T^3-T^2+T}{T^2-T+1} & 1 \\
 0 & 1 & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} &
   \frac{T}{T^2-T+1} & \frac{T^2}{T^2-T+1} & 1 \\
 0 & 0 & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} &
   \frac{T}{T^2-T+1} & \frac{T^2}{T^2-T+1} & 1 \\
 0 & 0 & \frac{1-T}{T^2-T+1} & \frac{1}{T^2-T+1} &
   \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & 1 \\
 0 & 0 & \frac{1-T}{T^2-T+1} & \frac{T-T^2}{T^2-T+1}
   & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & 1 \\
 0 & 0 & 0 & 0 & 0 & 1 & 1 \\
 0 & 0 & 0 & 0 & 0 & 0 & 1 \\
\end{array}
\right)
```

tex

It remains to blindly follow the two parts of Equation~\eqref{eq:rho1}:

pdf

In[∘]:= $\rho 1 = \sum_{k=1}^{n} R_1$ @@ Cs⟦k⟧ − $\sum_{k=1}^{2n} \varphi$⟦k⟧ $(g_{kk} - 1/2)$

pdf

Out[∘]= $-2 + g_{4,4} - g_{1,1}\,(-1 + g_{4,5}) + g_{2,5}\,(g_{2,3} + g_{3,2} - g_{5,2}) + g_{4,1}\,(-g_{1,4} + g_{4,5} + g_{5,4}) - (-1 + g_{2,3})\,g_{5,5} - g_{3,3}\,(-1 + g_{6,7}) + g_{6,3}\,(-g_{3,6} + g_{6,7} + g_{7,6})$

tex

We replace each ${\mathtt g}_{\alpha\beta}$ with the appropriate entry of \verb$G$:

In[∘]:= $\Delta^2\,\rho 1$ /. $g_{\alpha\_,\beta\_}$:→ G⟦α, β⟧

Out[∘]= $$\frac{\left(1 - T + T^2\right)^2 \left(-1 + \frac{T}{\left(1-T+T^2\right)^2} - \frac{-1+\frac{1}{1-T+T^2}}{1-T+T^2}\right)}{T^2}$$

tex

Finally, we output both Δ and ρ_1. We factor them just to put them in a nicer form:

pdf

In[∘]:= Factor@{Δ, $\Delta^2\,\rho 1$ /. $g_{\alpha\_,\beta\_}$:→ G⟦α, β⟧}

pdf

Out[∘]= $$\left\{\frac{1 - T + T^2}{T}, -\frac{(-1 + T)^2\,\left(1 + T^2\right)}{T^2}\right\}$$

tex

\subsection{A Demo Run} Here are Δ and ρ_1 of all the knots with up to 6 crossings:

pdf

In[∘]:=
```
Do[Echo[K → ρ[K]], {K, AllKnots[{3, 6}]}]
```

pdf

KnotTheory: Loading precomputed data in PD4Knots`.

pdf

» $\text{Knot}[3,1] \to \left\{\frac{1-T+T^2}{T}, \frac{(-1+T)^2\,(1+T^2)}{T^2}\right\}$

pdf

» $\text{Knot}[4,1] \to \left\{-\frac{1-3T+T^2}{T}, 0\right\}$

pdf

» $\text{Knot}[5,1] \to \left\{\frac{1-T+T^2-T^3+T^4}{T^2}, \frac{(-1+T)^2\,(1+T^2)\,(2+T^2+2T^4)}{T^4}\right\}$

pdf

» $\text{Knot}[5,2] \to \left\{\frac{2-3T+2T^2}{T}, \frac{(-1+T)^2\,(5-4T+5T^2)}{T^2}\right\}$

pdf

» $\text{Knot}[6,1] \to \left\{-\frac{(-2+T)\,(-1+2T)}{T}, \frac{(-1+T)^2\,(1-4T+T^2)}{T^2}\right\}$

pdf

» $\text{Knot}[6,2] \to \left\{-\frac{1-3T+3T^2-3T^3+T^4}{T^2}, \frac{(-1+T)^2\,(1-4T+4T^2-4T^3+4T^4-4T^5+T^6)}{T^4}\right\}$

pdf

» $\text{Knot}[6,3] \to \left\{\frac{1-3T+5T^2-3T^3+T^4}{T^2}, 0\right\}$

tex

\begin{figure}
\[\resizebox{6in}{!}{\input{figs/GST48-Marked.pdf_t}} \]
\caption{A 48-crossing knot from~\cite{GompfScharlemannThompson:Counterexample}.}
\label{fig:GST48}
\end{figure}
Next is one of our favourites, a knot from~\cite{GompfScharlemannThompson:Counterexample} (see Figure~\ref{fig:GST48}), which is a potential counterexample to the ribbon$=$slice conjecture. It takes about two minutes to compute ρ_1 for this 48 crossing knot (note that Mathematica prints \verb$Timing$ information is seconds, and that this information is highly dependent on the CPU used, how loaded it is, and even on its temperature at the time of the computation):

pdf

In[ ]:= $\text{Timing@}\rho[\text{EPD}[X_{14,1}, \overline{X}_{2,29}, X_{3,40}, X_{43,4}, \overline{X}_{26,5}, X_{6,95}, X_{96,7}, X_{13,8}, \overline{X}_{9,28}, X_{10,41}, X_{42,11}, \overline{X}_{27,12}, X_{30,15}, \overline{X}_{16,61}, \overline{X}_{17,72}, \overline{X}_{18,83}, X_{19,34}, \overline{X}_{89,20}, \overline{X}_{21,92}, \overline{X}_{79,22}, \overline{X}_{68,23}, \overline{X}_{57,24}, \overline{X}_{25,56}, X_{62,31}, X_{73,32}, X_{84,33}, \overline{X}_{50,35}, X_{36,81}, X_{37,70}, X_{38,59}, \overline{X}_{39,54}, X_{44,55}, X_{58,45}, X_{69,46}, X_{80,47}, X_{48,91}, X_{90,49}, X_{51,82}, X_{52,71}, X_{53,60}, \overline{X}_{63,74}, \overline{X}_{64,85}, \overline{X}_{76,65}, \overline{X}_{87,66}, \overline{X}_{67,94}, \overline{X}_{75,86}, \overline{X}_{88,77}, \overline{X}_{78,93}]]$

pdf

Out[ ]= $$\left\{158.625, \left\{-\frac{(-1+2T-T^2-T^3+2T^4-T^5+T^8)\,(-1+T^3-2T^4+T^5+T^6-2T^7+T^8)}{T^8},\right.\right.$$
$$\frac{1}{T^{16}}\,(-1+T)^2\,(5-18T+33T^2-32T^3+2T^4+42T^5-62T^6-8T^7+166T^8-242T^9+108T^{10}+$$
$$132T^{11}-226T^{12}+148T^{13}-11T^{14}-36T^{15}-11T^{16}+148T^{17}-226T^{18}+132T^{19}+108T^{20}-$$
$$\left.\left.242T^{21}+166T^{22}-8T^{23}-62T^{24}+42T^{25}+2T^{26}-32T^{27}+33T^{28}-18T^{29}+5T^{30})\right\}\right\}$$

tex

\subsection{The Separation Power of ρ_1} Let us check how powerful is ρ_1 on knots with up to 12 crossings:

pdf

```
{NumberOfKnots[{3, 12}],
 Length@Union@Table[ρ[K], {K, AllKnots[{3, 12}]}],
 Length@Union@Table[{HOMFLYPT[K], Kh[K]}, {K, AllKnots[{3, 12}]}]}
```

pdf

Out[∘]= {2977, 2882, 2785}

tex

So the pair (Δ,ρ_1) attains 2,882 distinct values on the 2,977 prime knots with up to 12 crossings (a deficit of 95), whereas the pair $(H,Kh)=$ (HOMFLYPT polynomial, Khovanov Homology) attains only 2,785 distinct values on the same knots (a deficit of 192).

tex

In our spare time we computed all of these invariants on all the prime knots with up to 14 crossings. On these 59,937 knots the pair (Δ,ρ_1) attains 53,684 distinct values (a deficit of 6,253) whereas the pair (H,Kh) attains only 49,149 distinct values on the same knots (a deficit of 10,788).

tex

Hence the pair (Δ,ρ_1), computable in polynomial time by simple programs, seems stronger than the pair (H,Kh), which is more difficult to program and (for all we know) cannot be computed in polynomial time.

---

exec

```
nb2tex$TeXFileName = "Invariance-R3.tex";
```

pdf

In[∘]:=

$$\delta_{i\_,j\_} := \mathrm{If}[i === j, 1, 0];$$
$$\mathrm{gRules}_{s\_,i\_,j\_} := \{g_{i\beta\_} \mapsto \delta_{i\beta} + T^s g_{i+1,\beta} + (1 - T^s)\, g_{j+1,\beta},\ g_{j\beta\_} \mapsto \delta_{j\beta} + g_{j+1,\beta},$$
$$g_{\alpha\_,i} \mapsto T^{-s}\, (g_{\alpha,i+1} - \delta_{\alpha,i+1}),\ g_{\alpha\_j} \mapsto g_{\alpha,j+1} - (1 - T^s)\, g_{\alpha i} - \delta_{\alpha,j+1}\}$$

pdf

In[∘]:=

$$\{i^+, j^+, k^+\} = \{i+1, j+1, k+1\};$$
$$\mathrm{lhs} = \mathrm{Simplify}[$$
$$R_1[1, j, k] + R_1[1, i, k^+] + R_1[1, i^+, j^+] \mathbin{//.} \mathrm{gRules}_{1,j,k} \cup \mathrm{gRules}_{1,i,k^+} \cup \mathrm{gRules}_{1,i^+,j^+}]$$

pdf

Out[∘]=

$$-\frac{1}{2\,T^2}\Big(-2\,(-1+T)\,T\,g^2_{2+j,2+i} +$$
$$2\,g_{2+j,2+i}\,\big(T^2 + T^2\,g_{2+i,2+j} - 2\,T^2\,g_{2+j,2+j} + g_{2+k,2+i} - 2\,T\,g_{2+k,2+i} + T^2\,g_{2+k,2+i} - T\,g_{2+k,2+j} + T^2\,g_{2+k,2+j}\big) +$$
$$2\,g_{2+i,2+i}\,\big(-2\,T^2 + (-1+T)\,T\,g_{2+j,2+i} + T^2\,g_{2+j,2+j} - g_{2+k,2+i} + T\,g_{2+k,2+i} + T^2\,g_{2+k,2+k}\big) +$$
$$T\,\Big(3\,T - 2\,(-1+T)\,g^2_{2+k,2+i} + 2\,T\,g_{2+k,2+j} + 2\,T\,g_{2+j,2+k}\,g_{2+k,2+j} + 2\,g^2_{2+k,2+j} -$$
$$2\,T\,g^2_{2+k,2+j} + 2\,g_{2+j,2+j}\,\big((-1+T)\,g_{2+k,2+i} + (-1+T)\,g_{2+k,2+j} + T\,(-1+g_{2+k,2+k})\big) -$$
$$4\,T\,g_{2+k,2+j}\,g_{2+k,2+k} + 2\,g_{2+k,2+i}\,\big(T + T\,g_{2+i,2+k} - 2\,(-1+T)\,g_{2+k,2+j} - 2\,T\,g_{2+k,2+k}\big)\Big)\Big)$$

pdf

In[∘]:=
```
rhs = Simplify[
  R1[1, i, j] + R1[1, i⁺, k] + R1[1, j⁺, k⁺] //. gRules1,i,j ⋃ gRules1,i⁺,k ⋃ gRules1,j⁺,k⁺]
```

pdf

Out[∘]=
$$-\frac{1}{2\,T^2}\Big(-2\,(-1+T)\,T\,g_{2+j,2+i}^2+ 2\,g_{2+j,2+i}\left(T^2+T^2\,g_{2+i,2+j}-2\,T^2\,g_{2+j,2+j}+g_{2+k,2+i}-2\,T\,g_{2+k,2+i}+T^2\,g_{2+k,2+i}-T\,g_{2+k,2+j}+T^2\,g_{2+k,2+j}\right)+ 2\,g_{2+i,2+i}\left(-2\,T^2+(-1+T)\,T\,g_{2+j,2+i}+T^2\,g_{2+j,2+j}-g_{2+k,2+i}+T\,g_{2+k,2+i}+T^2\,g_{2+k,2+k}\right)+ T\Big(3\,T-2\,(-1+T)\,g_{2+k,2+i}^2+2\,T\,g_{2+k,2+j}+2\,T\,g_{2+j,2+k}\,g_{2+k,2+j}+2\,g_{2+k,2+j}^2- 2\,T\,g_{2+k,2+j}^2+2\,g_{2+j,2+j}\left((-1+T)\,g_{2+k,2+i}+(-1+T)\,g_{2+k,2+j}+T\,(-1+g_{2+k,2+k})\right)- 4\,T\,g_{2+k,2+j}\,g_{2+k,2+k}+2\,g_{2+k,2+i}\left(T+T\,g_{2+i,2+k}-2\,(-1+T)\,g_{2+k,2+j}-2\,T\,g_{2+k,2+k}\right)\Big)\Big)$$

pdf

In[∘]:=
```
lhs == rhs
```

pdf

Out[∘]= True

---

exec

```
nb2tex$TeXFileName = "Invariance-R2c.tex";
```

pdf

In[∘]:=
```
{i⁺, j⁺} = {i + 1, j + 1};
Simplify[R1[-1, i, j⁺] + R1[1, i⁺, j] - (gj⁺,j⁺ - 1 / 2)]
lhs = Simplify[R1[-1, i, j⁺] + R1[1, i⁺, j] - (gj⁺,j⁺ - 1 / 2) //. gRules-1,i,j⁺ ⋃ gRules1,i⁺,j]
```

pdf

Out[∘]=
$$\frac{1}{2}-g_{1+i,1+i}\left(-1+g_{j,1+j}\right)+g_{j,1+i}\left(-g_{1+i,j}+g_{j,1+j}+g_{1+j,j}\right)- g_{1+j,1+j}+g_{i,i}\left(-1+g_{1+j,2+j}\right)-g_{1+j,i}\left(-g_{i,1+j}+g_{1+j,2+j}+g_{2+j,1+j}\right)$$

pdf

Out[∘]=
$$\frac{1}{2}-g_{2+j,2+j}$$

In[∘]:=
```
rhs = -(gj+2,j+2 - 1 / 2)
```

Out[∘]=
$$\frac{1}{2}-g_{2+j,2+j}$$

In[∘]:=
```
Length[AllKnots[{3, 13}]]
```

Out[∘]= 12 965

```
In[∘]:= Monitor[Timing[Tallyρ13 = Tally[Last /@ Tally@Table[ρ[K], {K, AllKnots[{3, 13}]}]]], K]
```

KnotTheory: Loading precomputed data in DTCode4KnotsTo11`.

KnotTheory: The GaussCode to PD conversion was written by Siddarth Sankaran at the University of Toronto in the summer of 2005.

KnotTheory: Loading precomputed data in KnotTheory/12A.dts.

KnotTheory: Loading precomputed data in KnotTheory/12N.dts.

General: Further output of KnotTheory::loading will be suppressed during this calculation.

```
Out[∘]= {20 270.5, {{1, 11 140}, {2, 809}, {4, 33}, {3, 23}, {6, 1}}}

In[∘]:= Total[Times @@@ Rest[Tallyρ13]]

Out[∘]= 1825

In[∘]:= Monitor[Timing[TallyHKh13 = Tally[Last /@
          Tally@Table[{Kh[PD@K][q, t], HOMFLYPT[PD@K][a, z]}, {K, AllKnots[{3, 13}]}]]], K]

Out[∘]= {950., {{1, 9714}, {2, 1269}, {3, 150}, {4, 47}, {5, 10}, {6, 3}, {7, 1}}}

In[∘]:= Total[Times @@@ Rest[TallyHKh13]]

Out[∘]= 3251

In[∘]:= {NumberOfKnots[14, Alternating], NumberOfKnots[14, NonAlternating]}

Out[∘]= {19 536, 27 436}

In[∘]:= 12 965 + 19 536 + 27 436

Out[∘]= 59 937

In[∘]:= Monitor[Timing[Tallyρ14 = Tally[Last /@ Tally@Table[ρ[K], {K, AllKnots[{3, 14}]}]]], K]
```

KnotTheory: Loading precomputed data in KnotTheory/14A.dts.

KnotTheory: Loading precomputed data in KnotTheory/14N.dts.

```
Out[∘]= {207 320., {{1, 48 336}, {2, 4814}, {3, 217}, {4, 291}, {6, 19}, {5, 4}, {8, 3}}}

In[∘]:= Monitor[Timing[TallyHKh14 = Tally[Last /@
          Tally@Table[{Kh[PD@K][q, t], HOMFLYPT[PD@K][a, z]}, {K, AllKnots[{3, 14}]}]]], K]

Out[∘]= {6727.34, {{1, 40 661}, {2, 6969}, {3, 965},
    {5, 85}, {4, 411}, {6, 43}, {8, 6}, {10, 1}, {9, 1}, {7, 7}}}

In[∘]:= {Total[Times @@@ Rest[Tallyρ14]], Total[Times @@@ Rest[TallyHKh14]]}

Out[∘]= {11 601, 19 276}

In[∘]:= Total[Times @@@ {{1, 40 661}, {2, 6969}, {3, 965},
      {5, 85}, {4, 411}, {6, 43}, {8, 6}, {10, 1}, {9, 1}, {7, 7}}]

Out[∘]= 59 937

In[∘]:= Total[Last @@@ {{1, 48 336}, {2, 4814}, {3, 217}, {4, 291}, {6, 19}, {5, 4}, {8, 3}}]

Out[∘]= 53 684
```

In[∘]:= **59 937 − 53 684**

Out[∘]= 6253

In[∘]:=
```
Total[Last @@@ {{1, 40 661}, {2, 6969}, {3, 965},
    {5, 85}, {4, 411}, {6, 43}, {8, 6}, {10, 1}, {9, 1}, {7, 7}}]
```

Out[∘]= 49 149

In[∘]:= **59 937 − 49 149**

Out[∘]= 10 788